In this Maple file, we compute the evolution equations for the Painlevé 24equation using the compatibility equation of the Lax system. We also obtain the expression of the Lax matrices in the geometric gauge without apparent singularities.
The deformation operator is h*(alpha12 \partial_{t_{\infty^{(1)},2}}+ alpha22 \partial_{t_{\infty^{(2)},2}}+alpha11\partial_{t_{\infty^{(1)},1}}+ alpha21\partial_{t_{\infty^{(2)},1}} +Lt *partial_t))
For convenience the irregular times and monodromies are denoted t_{i,j} at lambda=infinity and s_{i,j} at lambda=t

Summary of previous files: We have the expression for some coefficients of the Lax matrix L and of A.

```
> restart:
  with(LinearAlgebra):
  P011:=s10+s20;
  P022:=s10*s20;
  Pinfty11:=-t12-t22;
  Pinfty01:=-t11-t21;
  Pinfty22:=t12*t22;
  Pinfty12:=t11*t22+t12*t21;
  Pinfty02:=t12*t20+t10*t22+t11*t21;
  CoherenceEquation:=t10+t20+s10+s20;

  P1:=x-> P011/(x-t)+Pinfty01+Pinfty11*x;
  P2:=x-> P022/(x-t)^2+P012/(x-t)+Pinfty02+Pinfty12*x+Pinfty22*x^2;
  tdP2:=unapply(P2(lambda)-P012/(lambda-t),lambda);

  c2bis:=(alpha12*t22-alpha22*t12)/(2*(t12-t22));
  c1bis:=(1/2)*(t12*t21-t11*t22)/(t12-t22)^2*(alpha12-alpha22)+
  (alpha11*t22-alpha21*t12)/(t12-t22);
  mubis:=(1/2)*(t*(t12-t22)-t11+t21)*(q-t)/(-t22+t12)^2*(alpha12-
  alpha22)+(alpha11-alpha21)*(q-t)/(t12-t22)+(q-t)*Lt/h;
  nuMinus1bis:=(alpha12-alpha22)/(2*(-t22+t12));
  nu0bis:=(1/2)*(t21-t11)/(t12-t22)^2*(alpha12-alpha22)+(alpha11-
  alpha21)/(t12-t22);


  dP1dlambda:=unapply(diff(P1(lambda),lambda),lambda):
  dP2dlambda:=unapply(diff(P2(lambda),lambda),lambda):
  L:=Matrix(2,2,0):
  L[1,1]:=0:
  L[1,2]:=1:
  L[2,1]:=-P2(lambda)+P012/(lambda-t) +C/(lambda-t) -h*t12 -h*p/
  (lambda-q):
  L[2,2]:= P1(lambda)-h/(lambda-t)+h/(lambda-q):

  C01:=C:
```

```
A:=Matrix(2,2,0):
A[1,1]:= c2*lambda^2+ c1*lambda +c0+rho/(lambda-q):
A[1,2]:=nuMinus1*lambda+nu0+mu/(lambda-q):
A[2,1]:= AA21(lambda):
A[2,2]:= AA22(lambda):
dAdlambda:=Matrix(2,2,0):
for i from 1 to 2 do for j from 1 to 2 do dAdlambda[i,j]:=diff(A
[i,j],lambda):  od: od:

L;
A;
Q2:=unapply(-p*(q-t),lambda):
J:=Matrix(2,2,0):
J[1,1]:=1:
J[1,2]:=0:
J[2,1]:=Q2(lambda)/(lambda-q):
J[2,2]:=(lambda-t)^1/(lambda-q):
J;
dJdlambda:=Matrix(2,2,0):
for i from 1 to 2 do for j from 1 to 2 do dJdlambda[i,j]:=diff(J
[i,j],lambda): od: od:
J:

LJ:=Matrix(2,2,0):
LJ[1,1]:=0:
LJ[1,2]:=0:
LJ[2,2]:=diff(J[2,2],q)*Lq+diff(J[2,2],p)*Lp+h*diff(J[2,2],t)*Lt:
LJ[2,1]:=diff(J[2,1],q)*Lq+diff(J[2,1],p)*Lp+h*diff(J[2,1],t)*Lt:
LJ:

checkL:=simplify(Multiply(Multiply(J,L),J^(-1))+h*Multiply
(dJdlambda,J^(-1))):
checkA:=simplify(Multiply(Multiply(J,A),J^(-1))+Multiply(LJ,J^
(-1))):
```

$$P011 := s10 + s20 \quad (1)$$

$$P022 := s10\, s20$$

$$Pinfty11 := -t12 - t22$$

$$Pinfty01 := -t11 - t21$$

$$Pinfty22 := t12\, t22$$

$$Pinfty12 := t11\, t22 + t12\, t21$$

$$Pinfty02 := t10\, t22 + t11\, t21 + t12\, t20$$

$$CoherenceEquation := t10 + t20 + s10 + s20$$

$$P1 := x \rightarrow \frac{P011}{x - t} + Pinfty01 + Pinfty11\, x$$

$$P2 := x \rightarrow \frac{P022}{(x-t)^2} + \frac{P012}{x-t} + Pinfty02 + Pinfty12\, x + Pinfty22\, x^2$$

$$tdP2 := \lambda \rightarrow \frac{s10\, s20}{(\lambda - t)^2} + t10\, t22 + t11\, t21 + t12\, t20 + (t11\, t22 + t12\, t21)\, \lambda + t12\, t22\, \lambda^2$$

$$c2bis := \frac{\alpha 12\, t22 - \alpha 22\, t12}{2\, t12 - 2\, t22}$$

$$c1bis := \frac{1}{2} \frac{(-t11\, t22 + t12\, t21)\,(\alpha 12 - \alpha 22)}{(t12 - t22)^2} + \frac{\alpha 11\, t22 - \alpha 21\, t12}{t12 - t22}$$

$$mubis := \frac{1}{2} \frac{(t\,(t12 - t22) - t11 + t21)\,(q - t)\,(\alpha 12 - \alpha 22)}{(t12 - t22)^2} + \frac{(\alpha 11 - \alpha 21)\,(q - t)}{t12 - t22} + \frac{(q - t)\, Lt}{h}$$

$$nuMinus1bis := \frac{\alpha 12 - \alpha 22}{2\, t12 - 2\, t22}$$

$$nu0bis := \frac{1}{2} \frac{(t21 - t11)\,(\alpha 12 - \alpha 22)}{(t12 - t22)^2} + \frac{\alpha 11 - \alpha 21}{t12 - t22}$$

$$\left[\left[0, 1\right], \left[-\frac{s10\, s20}{(\lambda - t)^2} - t10\, t22 - t11\, t21 - t12\, t20 - (t11\, t22 + t12\, t21)\, \lambda - t12\, t22\, \lambda^2 + \frac{C}{\lambda - t} - h\, t12 - \frac{h\, p}{\lambda - q}, \frac{s10 + s20}{\lambda - t} - t11 - t21 + (-t12 - t22)\, \lambda - \frac{h}{\lambda - t} + \frac{h}{\lambda - q}\right]\right]$$

$$\begin{bmatrix} c2\, \lambda^2 + c1\, \lambda + c0 + \frac{\rho}{\lambda - q} & nuMinus1\, \lambda + \nu 0 + \frac{\mu}{\lambda - q} \\ AA21(\lambda) & AA22(\lambda) \end{bmatrix}$$

$$\begin{bmatrix} 1 & 0 \\ -\frac{p\,(q - t)}{\lambda - q} & \frac{\lambda - t}{\lambda - q} \end{bmatrix}$$

Solving the compatibility equations to obtain the Hamiltonian evolutions.

The compatibility equation is \mathcal{L}L=h\partial_\lambda A+[A,L]
Since the first line of L is trivial, we may easily obtain A[2,1] et A[2,2] to obtain the full expression for A

```
> LL:=h*dAdlambda+(Multiply(A,L)-Multiply(L,A)):

  Entry11:=LL[1,1]:
  Entry12:=LL[1,2]:

  AA21:=unapply(solve(Entry11=0,AA21(lambda)),lambda):
```

```
AA21bis:=h*dAdlambda[1,1]+A[1,2]*L[2,1]:

simplify(AA21(lambda)-AA21bis);
AA22:=unapply(solve(Entry12=0,AA22(lambda)),lambda):
AA22bis:=h*dAdlambda[1,2]+A[1,1]+A[1,2]*L[2,2]:

simplify(AA22(lambda)-AA22bis);
simplify(Entry11);
simplify(Entry12);
```

$$0 \quad (1.1)$$
$$0$$
$$0$$
$$0$$

We now compute the action of \mathcal{L} on L[2,2] et L[2,1] to obtain the evolution equations

Evolution of entry L_{2,2}

```
> Entry22:=simplify(LL[2,2]);
Entry22TermLambdaMinusqCube:=factor(residue(Entry22*(lambda-q)
^2,lambda=q));
Entry22TermLambdaMinusqSquare:=factor(residue(Entry22*(lambda-
q),lambda=q));
Entry22TermLambdaMinusq:=factor(residue(Entry22,lambda=q));

Entry22TermLambdaInfty4:=factor(-residue(Entry22/lambda^5,
lambda=infinity));
Entry22TermLambdaInfty3:=factor(-residue(Entry22/lambda^4,
lambda=infinity));
Entry22TermLambdaInfty2:=factor(-residue(Entry22/lambda^3,
lambda=infinity));
Entry22TermLambdaInfty1:=factor(-residue(Entry22/lambda^2,
lambda=infinity));
Entry22TermLambdaInfty0:=factor(-residue(Entry22/lambda,lambda=
infinity));
Entry22TermLambdaInftyMinus1:=factor(-residue(Entry22/lambda^2,
lambda=infinity));
Entry22TermLambdaInftyMinus2:=factor(-residue(Entry22/lambda^3,
lambda=infinity));
Entry22TermLambdaTMinus1:=factor(residue(Entry22,lambda=t));
Entry22TermLambdaTMinus2:=factor(residue(Entry22*(lambda-t),
lambda=t));

simplify( Entry22-(Entry22TermLambdaMinusqSquare/(lambda-q)^2+
```

```
Entry22TermLambdaMinusq/(lambda-q)
+Entry22TermLambdaInfty0+Entry22TermLambdaInfty1*lambda+
Entry22TermLambdaInfty2*lambda^2+Entry22TermLambdaInfty3*
lambda^3+Entry22TermLambdaInfty4*
lambda^4+Entry22TermLambdaTMinus1/(lambda-t)
+Entry22TermLambdaTMinus2/(lambda-t)^2) );
L[2,2];
```

$$Entry22 := -\frac{1}{(\lambda-q)^2(\lambda-t)^2}\Big(\Big(\big((2\,t12+2\,t22)\,nuMinus1-4\,c2\big)\,\lambda^5+\Big(\big((-4\,t12-4\,t22)\,nuMinus1+8\,c2\big)\,t+\big((-4\,t12-4\,t22)\,nuMinus1+8\,c2\big)\,q+(t21+t11)\,nuMinus1+(t12+t22)\,\nu0-2\,c1\Big)\,\lambda^4+\Big(\big((2\,t12+2\,t22)\,nuMinus1-4\,c2\big)\,t^2+\Big(\big((8\,t12+8\,t22)\,nuMinus1-16\,c2\big)\,q+(-2\,t11-2\,t21)\,nuMinus1+(-2\,t12-2\,t22)\,\nu0+4\,c1\Big)\,t-2\,\Big(\big((-t12-t22)\,nuMinus1+2\,c2\big)\,q+(t21+t11)\,nuMinus1+(t12+t22)\,\nu0-2\,c1\Big)\,q\Big)\,\lambda^3+\Big(\Big(\big((-4\,t12-4\,t22)\,nuMinus1+8\,c2\big)\,q+(t21+t11)\,nuMinus1+(t12+t22)\,\nu0-2\,c1\Big)\,t^2+\Big(\big((-4\,t12-4\,t22)\,nuMinus1+8\,c2\big)\,q^2+\big((4\,t11+4\,t21)\,nuMinus1+(4\,t12+4\,t22)\,\nu0-8\,c1\big)\,q-nuMinus1\,(h-s20-s10)\Big)\,t+\Big((t21+t11)\,nuMinus1+(t12+t22)\,\nu0-2\,c1\Big)\,q^2+\big(h\,nuMinus1-\mu\,(t12+t22)\big)\,q+(s10+s20)\,\nu0+(-t11-t21)\,\mu+2\,\rho\Big)\,\lambda^2+\Big(-2\,\Big(\big((-t12-t22)\,nuMinus1+2\,c2\big)\,q+(t21+t11)\,nuMinus1+(t12+t22)\,\nu0-2\,c1\Big)\,q\,t^2+\Big(\big((-2\,t11-2\,t21)\,nuMinus1+(-2\,t12-2\,t22)\,\nu0+4\,c1\big)\,q^2+\big((-2\,s20-2\,s10)\,nuMinus1+2\,\mu\,(t12+t22)\big)\,q-2\,h\,\nu0+(2\,t11+2\,t21)\,\mu-4\,\rho\Big)\,t+2\,(\nu0\,q-\mu)\,(h-s20-s10)\Big)\,\lambda+\Big(\big((t21+t11)\,nuMinus1+(t12+t22)\,\nu0-2\,c1\big)\,q^2+\big(h\,nuMinus1-\mu\,(t12+t22)\big)\,q+h\,\nu0+(-t11-t21)\,\mu+2\,\rho\Big)\,t^2-(q^2\,nuMinus1-\mu)\,(h-s20-s10)\,t-q\,(\nu0\,q-\mu)\,(h-s20-s10)\Big)\,h\Big) \tag{1.2}$$

$$Entry22TermLambdaMinusqCube := 0$$

$$Entry22TermLambdaMinusqSquare := \frac{1}{q-t}\Big(\big(-h\,q^2\,nuMinus1+h\,q\,t\,nuMinus1+\mu\,q^2\,t12+\mu\,q^2\,t22-\mu\,q\,t\,t12-\mu\,q\,t\,t22-h\,\nu0\,q+h\,\nu0\,t+\mu\,q\,t11+\mu\,q\,t21-\mu\,t\,t11-\mu\,t\,t21+h\,\mu-\mu\,s10-\mu\,s20-2\,q\,\rho+2\,\rho\,t\big)\,h\Big)$$

$$Entry22TermLambdaMinusq := 0$$

$$Entry22TermLambdaInfty4 := 0$$

$$Entry22TermLambdaInfty3 := 0$$

$$Entry22TermLambdaInfty2 := 0$$

$$Entry22TermLambdaInfty1 := 2\,(-t12\,nuMinus1-t22\,nuMinus1+2\,c2)\,h$$

$$Entry22TermLambdaInfty0 := h\,(-\nu0\,t12-\nu0\,t22-t11\,nuMinus1-t21\,nuMinus1+2\,c1)$$

$$Entry22TermLambdaInftyMinus1 := 2\,(-t12\,nuMinus1-t22\,nuMinus1+2\,c2)\,h$$

$$Entry22TermLambdaInftyMinus2 := 0$$

$$Entry22TermLambdaTMinus1 := 0$$

$$Entry22TermLambdaTMinus2 := -\frac{h\left(-q\,t\,nuMinus1 + t^2\,nuMinus1 - \nu0\,q + \nu0\,t + \mu\right)(h - s20 - s10)}{q - t}$$

$$0$$

$$\frac{s10 + s20}{\lambda - t} - t11 - t21 + (-t12 - t22)\,\lambda - \frac{h}{\lambda - t} + \frac{h}{\lambda - q}$$

Since the deformation operator is \hbar (alpha12\partial_{t_{\infty^{(1)},2} +alpha22\partial_{t_{\infty^{(2)},2}+alpha11\partial_{t_{\infty^{(1)},1} +alpha21\partial_{t_{\infty^{(2)},1})) we can obtain L[q]

```
> nu0:=solve(Entry22TermLambdaTMinus2=(s10+s20-h)*Lt,nu0);
```

$$\nu0 := -\frac{h\,q\,t\,nuMinus1 - h\,t^2\,nuMinus1 + Lt\,q - Lt\,t - h\,\mu}{(q - t)\,h} \quad (1.3)$$

```
> L22OrderLambda2:=-residue(L[2,2]/lambda^3,lambda=infinity);
  L22OrderLambda1:=-residue(L[2,2]/lambda^2,lambda=infinity);
  L22OrderLambda0:=-residue(L[2,2]/lambda^1,lambda=infinity);
  L22OrderLambdaMinus1:=-residue(L[2,2]/lambda^2,lambda=infinity)
  ;
  L22OrderT1:=residue(L[2,2],lambda=t):
  factor(simplify(h*(alpha12*diff(L22OrderLambda2,t12)+alpha22*
  diff(L22OrderLambda2,t22)
  +alpha11*diff(L22OrderLambda2,t11)+alpha21*diff
  (L22OrderLambda2,t21)   +Lt*diff(L22OrderLambda2,t) )
  - Entry22TermLambdaInfty2));

  Equation1:=factor(simplify(h*(alpha12*diff(L22OrderLambda1,t12)
  +alpha22*diff(L22OrderLambda1,t22)
  +alpha11*diff(L22OrderLambda1,t11)+alpha21*diff
  (L22OrderLambda1,t21)+Lt*diff(L22OrderLambda1,t))
  - Entry22TermLambdaInfty1));

  Equation2:=factor(simplify(h*(alpha12*diff(L22OrderLambda0,t12)
  +alpha22*diff(L22OrderLambda0,t22)
  +alpha11*diff(L22OrderLambda0,t11)+alpha21*diff
  (L22OrderLambda0,t21)+Lt*diff(L22OrderLambda0,t))
  - Entry22TermLambdaInfty0));

  Equation3:=factor(simplify(h*(alpha12*diff
  (L22OrderLambdaMinus1,t12)+alpha22*diff(L22OrderLambdaMinus1,
  t22)
  +alpha11*diff(L22OrderLambdaMinus1,t11)+alpha21*diff
  (L22OrderLambdaMinus1,t21)+Lt*diff(L22OrderLambdaMinus1,t))
  - Entry22TermLambdaInftyMinus1));
```

```
Equation3bis:=factor(simplify(h*(alpha12*diff(L22OrderT1,t12)+
alpha22*diff(L22OrderT1,t22)
+alpha11*diff(L22OrderT1,t11)+alpha21*diff(L22OrderT1,t21)+
Lt*diff(L22OrderT1,t))
- Entry22TermLambdaTMinus1));
```

$$L22OrderLambda2 := 0 \quad \textbf{(1.4)}$$
$$L22OrderLambda1 := -t12 - t22$$
$$L22OrderLambda0 := -t11 - t21$$
$$L22OrderLambdaMinus1 := -t12 - t22$$
$$0$$
$$Equation1 := -h\,(-2\,t12\,nuMinus1 - 2\,t22\,nuMinus1 + \alpha 12 + \alpha 22 + 4\,c2)$$
$$Equation2 := -\frac{1}{q-t}\big(h\,q\,t\,t12\,nuMinus1 + h\,q\,t\,t22\,nuMinus1 - h\,t^2\,t12\,nuMinus1 - h\,t^2\,t22\,nuMinus1 - h\,q\,t11\,nuMinus1 - h\,q\,t21\,nuMinus1 + h\,t\,t11\,nuMinus1 + h\,t\,t21\,nuMinus1 + Lt\,q\,t12 + Lt\,q\,t22 - Lt\,t\,t12 - Lt\,t\,t22 + \alpha 11\,h\,q - \alpha 11\,h\,t + \alpha 21\,h\,q - \alpha 21\,h\,t + 2\,c1\,h\,q - 2\,c1\,h\,t - h\,\mu\,t12 - h\,\mu\,t22\big)$$
$$Equation3 := -h\,(-2\,t12\,nuMinus1 - 2\,t22\,nuMinus1 + \alpha 12 + \alpha 22 + 4\,c2)$$
$$Equation3bis := 0$$

```
> Lq:=factor(Entry22TermLambdaMinusqSquare/h):
Lqbis:=-2*rho-mu*q*P1(q)-(1/2)*h*q/(t12-t22)*(alpha12-alpha22):
```

We now look at \mathcal{L}[L[2,1]]

```
> Entry21:=simplify(LL[2,1]):
Entry21TermLambdaMinusqCube:=factor(residue(Entry21*(lambda-q)
^2,lambda=q));
Entry21TermLambdaMinusqSquare:=factor(residue(Entry21*(lambda-
q),lambda=q));
Entry21TermLambdaMinusq:=factor(residue(Entry21,lambda=q));
Entry21TermLambdaInfty2:=factor(-residue(Entry21/lambda^3,
lambda=infinity));
Entry21TermLambdaInfty1:=factor(-residue(Entry21/lambda^2,
lambda=infinity));
Entry21TermLambdaInfty0:=factor(-residue(Entry21/lambda,lambda=
infinity));
Entry21TermLambdaT1:=factor(residue(Entry21,lambda=t));
Entry21TermLambdaT2:=factor(residue(Entry21*(lambda-t),lambda=
t));

simplify( Entry21-(Entry21TermLambdaMinusqCube/(lambda-q)^3+
Entry21TermLambdaMinusqSquare/(lambda-q)^2+
Entry21TermLambdaMinusq/(lambda-q)
+Entry21TermLambdaInfty0+Entry21TermLambdaInfty1*
```

```
lambda+Entry21TermLambdaInfty2*lambda^2
+Entry21TermLambdaT1/(lambda-t)
) );
L[2,1];
```

$$Entry21TermLambdaMinusqCube := 3\,(\mu p+\rho)\,h^2 \qquad (1.5)$$

$$Entry21TermLambdaMinusqSquare := -\frac{1}{(q-t)^2}\big((-2\,\mu\,q^4\,t12\,t22+4\,\mu\,q^3\,t\,t12\,t22 -2\,\mu\,q^2\,t^2\,t12\,t22-h\,p\,q^3\,nuMinus1+3\,h\,p\,q^2\,t\,nuMinus1-3\,h\,p\,q\,t^2\,nuMinus1 +h\,p\,t^3\,nuMinus1-2\,\mu\,q^3\,t11\,t22-2\,\mu\,q^3\,t12\,t21+4\,\mu\,q^2\,t\,t11\,t22+4\,\mu\,q^2\,t\,t12\,t21 -2\,\mu\,q\,t^2\,t11\,t22-2\,\mu\,q\,t^2\,t12\,t21-2\,h\,\mu\,q^2\,t12+4\,h\,\mu\,q\,t\,t12-2\,h\,\mu\,t^2\,t12 -2\,\mu\,q^2\,t10\,t22-2\,\mu\,q^2\,t11\,t21-2\,\mu\,q^2\,t12\,t20+4\,\mu\,q\,t\,t10\,t22+4\,\mu\,q\,t\,t11\,t21 +4\,\mu\,q\,t\,t12\,t20-2\,\mu\,t^2\,t10\,t22-2\,\mu\,t^2\,t11\,t21-2\,\mu\,t^2\,t12\,t20+q^3\,\rho\,t12+q^3\,\rho\,t22 -2\,q^2\,\rho\,t\,t12-2\,q^2\,\rho\,t\,t22+q\,\rho\,t^2\,t12+q\,\rho\,t^2\,t22+Lt\,p\,q^2-2\,Lt\,p\,q\,t+Lt\,p\,t^2 -h\,\mu\,p\,q+h\,\mu\,p\,t+q^2\,\rho\,t11+q^2\,\rho\,t21-2\,q\,\rho\,t\,t11-2\,q\,\rho\,t\,t21+\rho\,t^2\,t11+\rho\,t^2\,t21 +2\,C\,\mu\,q-2\,C\,\mu\,t+h\,q\,\rho-h\,\rho\,t-2\,\mu\,s10\,s20-q\,\rho\,s10-q\,\rho\,s20+\rho\,s10\,t +\rho\,s20\,t)\,h\big)$$

$$Entry21TermLambdaMinusq := \frac{1}{(q-t)^3}\big(h\,(2\,\mu\,q^4\,t12\,t22-6\,\mu\,q^3\,t\,t12\,t22 +6\,\mu\,q^2\,t^2\,t12\,t22-2\,\mu\,q\,t^3\,t12\,t22-2\,c2\,h\,q^4+6\,c2\,h\,q^3\,t-6\,c2\,h\,q^2\,t^2+2\,c2\,h\,q\,t^3 -h\,p\,q^3\,nuMinus1+3\,h\,p\,q^2\,t\,nuMinus1-3\,h\,p\,q\,t^2\,nuMinus1+h\,p\,t^3\,nuMinus1 +\mu\,q^3\,t11\,t22+\mu\,q^3\,t12\,t21-3\,\mu\,q^2\,t\,t11\,t22-3\,\mu\,q^2\,t\,t12\,t21+3\,\mu\,q\,t^2\,t11\,t22 +3\,\mu\,q\,t^2\,t12\,t21-\mu\,t^3\,t11\,t22-\mu\,t^3\,t12\,t21-c1\,h\,q^3+3\,c1\,h\,q^2\,t-3\,c1\,h\,q\,t^2 +c1\,h\,t^3-q^3\,\rho\,t12-q^3\,\rho\,t22+3\,q^2\,\rho\,t\,t12+3\,q^2\,\rho\,t\,t22-3\,q\,\rho\,t^2\,t12-3\,q\,\rho\,t^2\,t22 +\rho\,t^3\,t12+\rho\,t^3\,t22+C\,\mu\,q-C\,\mu\,t+h\,q\,\rho-h\,\rho\,t-2\,\mu\,s10\,s20-q\,\rho\,s10-q\,\rho\,s20 +\rho\,s10\,t+\rho\,s20\,t)\big)$$

$$Entry21TermLambdaInfty2 := 2\,(-2\,t12\,t22\,nuMinus1+c2\,t12+c2\,t22)\,h$$

$$Entry21TermLambdaInfty1 := \frac{1}{q-t}\big(2\,h\,q\,t\,t12\,t22\,nuMinus1-2\,h\,t^2\,t12\,t22\,nuMinus1 -3\,h\,q\,t11\,t22\,nuMinus1-3\,h\,q\,t12\,t21\,nuMinus1+3\,h\,t\,t11\,t22\,nuMinus1 +3\,h\,t\,t12\,t21\,nuMinus1+2\,Lt\,q\,t12\,t22-2\,Lt\,t\,t12\,t22+c1\,h\,q\,t12+c1\,h\,q\,t22 -c1\,h\,t\,t12-c1\,h\,t\,t22+2\,c2\,h\,q\,t11+2\,c2\,h\,q\,t21-2\,c2\,h\,t\,t11-2\,c2\,h\,t\,t21 -2\,h\,\mu\,t12\,t22\big)$$

$$Entry21TermLambdaInfty0 := \frac{1}{q-t}\big(h\,q\,t\,t11\,t22\,nuMinus1+h\,q\,t\,t12\,t21\,nuMinus1 -h\,t^2\,t11\,t22\,nuMinus1-h\,t^2\,t12\,t21\,nuMinus1-2\,h^2\,q\,t12\,nuMinus1 +2\,h^2\,t\,t12\,nuMinus1-2\,h\,q\,t10\,t22\,nuMinus1-2\,h\,q\,t11\,t21\,nuMinus1 -2\,h\,q\,t12\,t20\,nuMinus1+2\,h\,t\,t10\,t22\,nuMinus1+2\,h\,t\,t11\,t21\,nuMinus1 +2\,h\,t\,t12\,t20\,nuMinus1+Lt\,q\,t11\,t22+Lt\,q\,t12\,t21-Lt\,t\,t11\,t22-Lt\,t\,t12\,t21 +c1\,h\,q\,t11+c1\,h\,q\,t21-c1\,h\,t\,t11-c1\,h\,t\,t21+2\,c2\,h^2\,q-2\,c2\,h^2\,t-2\,c2\,h\,q\,s10$$

$- 2\, c2\, h\, q\, s20 + 2\, c2\, h\, s10\, t + 2\, c2\, h\, s20\, t - h\, \mu\, t11\, t22 - h\, \mu\, t12\, t21)$

$$Entry21TermLambdaT1 := -\frac{1}{(q-t)^3} \big(\big(-2\, c2\, h\, q^3\, t + 6\, c2\, h\, q^2\, t^2 - 6\, c2\, h\, q\, t^3 + 2\, c2\, h\, t^4 + 2\, c2\, q^3\, s10\, t + 2\, c2\, q^3\, s20\, t - 6\, c2\, q^2\, s10\, t^2 - 6\, c2\, q^2\, s20\, t^2 + 6\, c2\, q\, s10\, t^3 + 6\, c2\, q\, s20\, t^3 - 2\, c2\, s10\, t^4 - 2\, c2\, s20\, t^4 - C\, q^3\, nuMinus1 + 3\, C\, q^2\, t\, nuMinus1 - 3\, C\, q\, t^2\, nuMinus1 + C\, t^3\, nuMinus1 - c1\, h\, q^3 + 3\, c1\, h\, q^2\, t - 3\, c1\, h\, q\, t^2 + c1\, h\, t^3 + c1\, q^3\, s10 + c1\, q^3\, s20 - 3\, c1\, q^2\, s10\, t - 3\, c1\, q^2\, s20\, t + 3\, c1\, q\, s10\, t^2 + 3\, c1\, q\, s20\, t^2 - c1\, s10\, t^3 - c1\, s20\, t^3 + C\, \mu\, q - C\, \mu\, t + h\, q\, \rho - h\, \rho\, t - 2\, \mu\, s10\, s20 - q\, \rho\, s10 - q\, \rho\, s20 + \rho\, s10\, t + \rho\, s20\, t \big)\, h \big)$$

$$Entry21TermLambdaT2 := C\, Lt$$

$$\frac{\big((t-\lambda)\, C + 2\, s10\, s20 \big)\, Lt}{(t-\lambda)^3}$$

$$-\frac{s10\, s20}{(\lambda - t)^2} - t10\, t22 - t11\, t21 - t12\, t20 - (t11\, t22 + t12\, t21)\, \lambda - t12\, t22\, \lambda^2 + \frac{C}{\lambda - t} - h\, t12 - \frac{h\, p}{\lambda - q}$$

```
> rho:=factor(solve(Entry21TermLambdaMinusqCube,rho));
  simplify(rho-(-p*q*mu));
  simplify(Entry21TermLambdaMinusqCube);
```

$$\rho := -p\, \mu \qquad \textbf{(1.6)}$$

$$p\, \mu\, (q-1)$$

$$0$$

```
> L21OrderLambda3:=-residue(L[2,1]/lambda^4,lambda=infinity);
  L21OrderLambda2:=-residue(L[2,1]/lambda^3,lambda=infinity);
  L21OrderLambda1:=-residue(L[2,1]/lambda^2,lambda=infinity);
  L21OrderLambda0:=-residue(L[2,1]/lambda^1,lambda=infinity);
  L21OrderLambdaMinus1:=-residue(L[2,1]/lambda^2,lambda=infinity)
  ;
  L21OrderLambdaMinus2:=-residue(L[2,1]/lambda^3,lambda=infinity)
  ;
  L21TOrder2:=factor(residue(L[2,1]*(lambda-t),lambda=t));
  L21TOrder1:=residue(L[2,1],lambda=t);
  Equation4:=simplify(h*(alpha12*diff(L21OrderLambda2,t12)+
  alpha22*diff(L21OrderLambda2,t22)+alpha11*diff(L21OrderLambda2,
  t11)+alpha21*diff(L21OrderLambda2,t21)+Lt*diff(L21OrderLambda2,
  t))- Entry21TermLambdaInfty2);
  Equation5:=simplify(h*(alpha12*diff(L21OrderLambda1,t12)+
  alpha22*diff(L21OrderLambda1,t22)+alpha11*diff(L21OrderLambda1,
  t11)+alpha21*diff(L21OrderLambda1,t21)+Lt*diff(L21OrderLambda1,
  t))- Entry21TermLambdaInfty1);
  Equation6:=simplify(h*(alpha12*diff(L21TOrder2,t12)+alpha22*
```

```
diff(L21TOrder2,t22)+alpha11*diff(L21TOrder2,t11)+alpha21*diff
(L21TOrder2,t21)+Lt*diff(L21TOrder2,t))- Entry21TermLambdaT2);
```

$$L21OrderLambda3 := 0 \tag{1.7}$$
$$L21OrderLambda2 := -t12\,t22$$
$$L21OrderLambda1 := -t11\,t22 - t12\,t21$$
$$L21OrderLambda0 := -h\,t12 - t10\,t22 - t11\,t21 - t12\,t20$$
$$L21OrderLambdaMinus1 := -t11\,t22 - t12\,t21$$
$$L21OrderLambdaMinus2 := -t12\,t22$$
$$L21TOrder2 := -s10\,s20$$
$$L21TOrder1 := \frac{-C\,q + C\,t}{-q+t}$$
$$Equation4 := -2\left(\left(-2\,t22\,numMinus1 + c2 + \frac{1}{2}\,\alpha22\right)t12 + t22\left(c2 + \frac{1}{2}\,\alpha12\right)\right)h$$
$$Equation5 := \frac{1}{q-t}\Big(\big(2\,t22\,numMinus1\,t12\,t^2 + \big(-2\,t12\,t22\,numMinus1\,q + \big(-3\,t21\,numMinus1 + \alpha21 + c1\big)\,t12 + \big(-3\,t11\,numMinus1 + \alpha11 + c1\big)\,t22 + \big(\alpha22 + 2\,c2\big)\,t11 + t21\,\big(\alpha12 + 2\,c2\big)\big)\,t + \big(\big(3\,t21\,numMinus1 - \alpha21 - c1\big)\,t12 + \big(3\,t11\,numMinus1 - \alpha11 - c1\big)\,t22 + \big(-\alpha22 - 2\,c2\big)\,t11 - t21\,\big(\alpha12 + 2\,c2\big)\big)\,q + 2\,t12\,t22\,\mu\big)\,h - 2\,t12\,t22\,Lt\,(q-t)\Big)$$
$$Equation6 := -C\,Lt$$

```
> c2:=factor(solve(Equation1,c2));
mu:=factor(solve(Equation2,mu));
numMinus1:=factor(solve(Equation4,numMinus1));
c1:=factor(solve(Equation5,c1));
c2:=simplify(c2);
mu:=factor(mu);
```

$$c2 := \frac{1}{2}\,t12\,numMinus1 + \frac{1}{2}\,t22\,numMinus1 - \frac{1}{4}\,\alpha12 - \frac{1}{4}\,\alpha22 \tag{1.8}$$
$$\mu := \frac{1}{h\,(t12+t22)}\Big(\big(h\,t\,t12\,numMinus1 + h\,t\,t22\,numMinus1 - h\,t11\,numMinus1 - h\,t21\,numMinus1 + Lt\,t12 + Lt\,t22 + \alpha11\,h + \alpha21\,h + 2\,c1\,h\big)\,(q-t)\Big)$$
$$numMinus1 := \frac{1}{2}\,\frac{\alpha12 - \alpha22}{t12 - t22}$$
$$c1 := \frac{1}{2}\,\frac{1}{(t12-t22)^2}\big(2\,\alpha11\,t12\,t22 - 2\,\alpha11\,t22^2 - \alpha12\,t11\,t22 + \alpha12\,t12\,t21 - 2\,\alpha21\,t12^2 + 2\,\alpha21\,t12\,t22 + \alpha22\,t11\,t22 - \alpha22\,t12\,t21\big)$$
$$c2 := \frac{\alpha12\,t22 - \alpha22\,t12}{2\,t12 - 2\,t22}$$
$$\mu := \frac{1}{2}\,\frac{1}{h\,(t12-t22)^2}\Big((q-t)\,\big(\alpha12\,h\,t\,t12 - \alpha12\,h\,t\,t22 - \alpha22\,h\,t\,t12 + \alpha22\,h\,t\,t22 + 2\,Lt\,t12^2 - 4\,Lt\,t12\,t22 + 2\,Lt\,t22^2 + 2\,\alpha11\,h\,t12 - 2\,\alpha11\,h\,t22 - \alpha12\,h\,t11 + \alpha12\,h\,t21 - 2\,\alpha21\,h\,t12 + 2\,\alpha21\,h\,t22 + \alpha22\,h\,t11 - \alpha22\,h\,t21\big)\Big)$$

```
> simplify(Equation1);
  simplify(Equation2);
  simplify(Equation3);
  simplify(Equation4);
  simplify(Equation5);
  simplify(c1-c1bis);
  simplify(c2-c2bis);
  simplify(mu-mubis);
  simplify(nu0-nu0bis);
  simplify(nuMinus1-nuMinus1bis);
```

$$\begin{gathered}0\\0\\0\\0\\0\\0\\0\\0\\0\\0\end{gathered} \tag{1.9}$$

```
> simplify(Entry21TermLambdaMinusqSquare-(-p*h*Lq)):
> Lpfunction:=unapply(-Entry21TermLambdaMinusq/h,C):
> Equation7:=unapply( simplify(Entry21TermLambdaMinusqSquare-(-p*
  h*Lq)),C):
  Cter:=(q^4*t12*t22-2*q^3*t*t12*t22+q^2*t^2*t12*t22+p*q^3*t12+p*
  q^3*t22-2*p*q^2*t*t12-2*p*q^2*t*t22+p*q*t^2*t12+p*q*t^2*t22+
  q^3*t11*t22+q^3*t12*t21-2*q^2*t*t11*t22-2*q^2*t*t12*t21+q*t^2*
  t11*t22+q*t^2*t12*t21+h*q^2*t12-2*h*q*t*t12+h*t^2*t12+p^2*q^2
  -2*p^2*q*t+p^2*t^2+p*q^2*t11+p*q^2*t21-2*p*q*t*t11-2*p*q*t*t21+
  p*t^2*t11+p*t^2*t21+q^2*t10*t22+q^2*t11*t21+q^2*t12*t20-2*q*t*
  t10*t22-2*q*t*t11*t21-2*q*t*t12*t20+t^2*t10*t22+t^2*t11*t21+
  t^2*t12*t20+h*p*q-h*p*t-p*q*s10-p*q*s20+p*s10*t+p*s20*t+s10*
  s20)/(q-t):
  simplify(Equation7(Cter));
  solve(Equation7(C),C):

  Cbis:=(q-t)*p^2+h*p-(q-t)*P1(q)*p+(q-t)*P2(q)-P012+h*t12*(q-t);
  simplify(Equation7(Cbis));
  simplify(series(Cter-Cbis,p));
  C:=Cbis:
```

$$0 \tag{1.10}$$

$$Cbis := (q-t)\,p^2 + h\,p - (q-t)\left(\frac{s10+s20}{q-t} - t11 - t21 + (-t12 - t22)\,q\right)p + (q$$

$$-t)\left(\frac{s10\,s20}{(q-t)^2}+\frac{P012}{q-t}+t10\,t22+t11\,t21+t12\,t20+(t11\,t22+t12\,t21)\,q+t12\,t22\,q^2\right)-P012+h\,t12\,(q-t)$$

$$0$$
$$0$$

```
> Lp:=factor(simplify(Lpfunction(Cbis))):
  Lpbis:=mu*(p*diff(P1(q),q)+p*h*1/(q-t)^2-diff(tdP2(q),q)- C01/
  (q-t)^2)+h*nuMinus1*p +h*c1+2*h*c2*q:
  factor(series(Lp-Lpbis,P012=0));
```

$$0 \tag{1.11}$$

```
> Lqter:=2*mu*(p-P1(q)/2+1/2*h*1/(q-t))-h*nu0-h*nuMinus1*q:
  simplify(Lq-Lqter);
```

$$0 \tag{1.12}$$

```
> nuMinus1:=nuMinus1;
  nu0:=nu0;
  c1:=c1;
  c2:=c2;
```

$$nuMinus1 := \frac{1}{2}\,\frac{\alpha 12-\alpha 22}{t12-t22} \tag{1.13}$$

$$\nu 0 := -\frac{1}{(q-t)\,h}\left(\frac{1}{2}\,\frac{h\,q\,t\,(\alpha 12-\alpha 22)}{t12-t22}-\frac{1}{2}\,\frac{h\,t^2\,(\alpha 12-\alpha 22)}{t12-t22}+Lt\,q-Lt\,t-\frac{1}{2}\,\frac{1}{(t12-t22)^2}\big((q-t)\,(\alpha 12\,h\,t\,t12-\alpha 12\,h\,t\,t22-\alpha 22\,h\,t\,t12+\alpha 22\,h\,t\,t22+2\,Lt\,t12^2-4\,Lt\,t12\,t22+2\,Lt\,t22^2+2\,\alpha 11\,h\,t12-2\,\alpha 11\,h\,t22-\alpha 12\,h\,t11+\alpha 12\,h\,t21-2\,\alpha 21\,h\,t12+2\,\alpha 21\,h\,t22+\alpha 22\,h\,t11-\alpha 22\,h\,t21)\big)\right)$$

$$c1 := \frac{1}{2}\,\frac{1}{(t12-t22)^2}\,(2\,\alpha 11\,t12\,t22-2\,\alpha 11\,t22^2-\alpha 12\,t11\,t22+\alpha 12\,t12\,t21-2\,\alpha 21\,t12^2+2\,\alpha 21\,t12\,t22+\alpha 22\,t11\,t22-\alpha 22\,t12\,t21)$$

$$c2 := \frac{\alpha 12\,t22-\alpha 22\,t12}{2\,t12-2\,t22}$$

We thus get that
L[q]=2*mu*q*(p-P1(q)/2)-(1/2)*h*q/(t12-t22)*(alpha12-alpha22) -h*mu
L[p] = -mu*p^2+mu*diff(p*q*P1(q)-q*P2(q),q)-h*t12*mu +1/2*h*(alpha12-alpha22)/(t12-t22)*p+h*c1+2*h*c2*q
with

$$\boldsymbol{nuMinus1 := \frac{\alpha 12-\alpha 22}{2\,(t12-t22)}}$$

$$\boldsymbol{\nu 0 := -\frac{1}{(q-t)\,h}\left(\frac{h\,q\,t\,(\alpha 12-\alpha 22)}{2\,(t12-t22)}-\frac{h\,t^2\,(\alpha 12-\alpha 22)}{2\,(t12-t22)}+Lt\,q-Lt\,t\right.}$$

$$-\frac{1}{2(t12-t22)^2}\Big((q-t)\big(\alpha 12\,h\,t\,t12-\alpha 12\,h\,t\,t22-\alpha 22\,h\,t\,t12+\alpha 22\,h\,t\,t22+2\,Lt\,t12^2-4\,Lt\,t12\,t22+2\,t22^2\,Lt+2\,\alpha 11\,h\,t12-2\,\alpha 11\,h\,t22-\alpha 12\,h\,t11+\alpha 12\,h\,t21-2\,\alpha 21\,h\,t12+2\,\alpha 21\,h\,t22+\alpha 22\,h\,t11-\alpha 22\,h\,t21\big)\Big)\Big)$$

$$c1:=\frac{1}{2(t12-t22)^2}\big(2\,t12\,\alpha 11\,t22-2\,t22^2\,\alpha 11-t11\,\alpha 12\,t22+t21\,\alpha 12\,t12-2\,t12^2\,\alpha 21+2\,t12\,\alpha 21\,t22+t11\,\alpha 22\,t22-t21\,\alpha 22\,t12\big)$$

$$c2:=\frac{\alpha 12\,t22-\alpha 22\,t12}{2\,t12-2\,t22}$$

```
> Hamiltonian:= mu*(p^2-P1(q)*p+P2(q))-1/2*h*(alpha12-alpha22)/
(t12-t22)*p*q+h*t12*mu -h*c1*q-h*c2*q^2+p*((2*Lt*t12-2*Lt*t22+
h*t*(alpha12-alpha22))/(2*t12-2*t22)):
simplify(Lp-(-diff(Hamiltonian,q)));
simplify(Lq-(diff(Hamiltonian,p)));

Hamiltonianbis:= mu*(p^2-P1(q)*p+h*p*1/(q-t) +tdP2(q)+h*t12 )-
h*nu0*p-h*nuMinus1*q*p-h*c1*q-h*c2*q^2  :
factor(simplify(Lp-(-diff(Hamiltonianbis,q))));
simplify(Lq-(diff(Hamiltonianbis,p)));
```

$$0 \qquad (1.14)$$
$$0$$
$$0$$
$$0$$

Expression of the Lax matrix in the geometric gauge and normalisation at infinity

```
> simplify(checkL[1,1]);
simplify(checkL[1,2]);
checkL22bis:=P1(lambda)-p*(q-t)/(lambda-t);
simplify(checkL[2,2]-checkL22bis);
checkL21:=factor(checkL[2,1]);
simplify(series(checkL[2,1],lambda=t)):
checkL21bis:=(p*(q-t)-s10)*(p*(q-t)-s20)/((q-t)*(lambda-t))-
t12*t22*lambda^2+(((-q+t)*t22-t21)*t12-t11*t22)*lambda+((-q^2+
q*t)*t22+(-t21-p)*q+(t21+p)*t-h-t20)*t12+((-p-t11)*q+(t11+p)*t-
t10)*t22-t11*t21:
simplify(checkL21-checkL21bis);
```

$$\frac{p\,(q-t)}{\lambda-t} \qquad (2.1)$$
$$\frac{-\lambda+q}{t-\lambda}$$

$$checkL22bis := \frac{s10+s20}{\lambda-t} - t11 - t21 + (-t12-t22)\,\lambda - \frac{p\,(q-t)}{\lambda-t}$$

$$0$$

$$checkL21 := -\frac{1}{(\lambda-t)\,(q-t)}\big(\lambda^3 q\,t12\,t22 - \lambda^3 t\,t12\,t22 + \lambda^2 q^2\,t12\,t22 - 3\,\lambda^2 q\,t\,t12\,t22 + 2\,\lambda^2 t^2\,t12\,t22 + \lambda q^3\,t12\,t22 - 3\,\lambda q^2 t\,t12\,t22 + 3\,\lambda q\,t^2\,t12\,t22 - \lambda t^3\,t12\,t22 - q^3 t\,t12\,t22 + 2\,q^2 t^2\,t12\,t22 - q\,t^3\,t12\,t22 + \lambda^2 q\,t11\,t22 + \lambda^2 q\,t12\,t21 - \lambda^2 t\,t11\,t22 - \lambda^2 t\,t12\,t21 + \lambda p\,q^2\,t12 + \lambda p\,q^2\,t22 - 2\,\lambda p\,q\,t\,t12 - 2\,\lambda p\,q\,t\,t22 + \lambda p\,t^2\,t12 + \lambda p\,t^2\,t22 + \lambda q^2\,t11\,t22 + \lambda q^2\,t12\,t21 - 3\,\lambda q\,t\,t11\,t22 - 3\,\lambda q\,t\,t12\,t21 + 2\,\lambda t^2\,t11\,t22 + 2\,\lambda t^2\,t12\,t21 - p\,q^2 t\,t12 - p\,q^2 t\,t22 + 2\,p\,q\,t^2\,t12 + 2\,p\,q\,t^2\,t22 - p\,t^3\,t12 - p\,t^3\,t22 - q^2 t\,t11\,t22 - q^2 t\,t12\,t21 + 2\,q\,t^2\,t11\,t22 + 2\,q\,t^2\,t12\,t21 - t^3\,t11\,t22 - t^3\,t12\,t21 + h\,\lambda q\,t12 - h\,\lambda t\,t12 - h\,q\,t\,t12 + h\,t^2\,t12 + \lambda q\,t10\,t22 + \lambda q\,t11\,t21 + \lambda q\,t12\,t20 - \lambda t\,t10\,t22 - \lambda t\,t11\,t21 - \lambda t\,t12\,t20 - p^2 q^2 + 2\,p^2 q\,t - p^2 t^2 - q\,t\,t10\,t22 - q\,t\,t11\,t21 - q\,t\,t12\,t20 + t^2\,t10\,t22 + t^2\,t11\,t21 + t^2\,t12\,t20 + p\,q\,s10 + p\,q\,s20 - p\,s10\,t - p\,s20\,t - s10\,s20\big)$$

$$0$$

```
> G1:=Matrix(2,2,0):
G1[1,1]:=1:
G1[2,2]:=1:
G1[1,2]:=0:
G1[2,1]:=t12*lambda+eta0:
g1:=t12:
eta0:=(q-t)*t12+t11;

dG1dlambda:=Matrix(2,2,0):
for i from 1 to 2 do for j from 1 to 2 do dG1dlambda[i,j]:=diff
(G1[i,j],lambda): od: od:

tdL:=simplify(Multiply(Multiply(G1,checkL),G1^(-1))+h*Multiply
(dG1dlambda,G1^(-1))):

simplify(tdL);
series(tdL[1,1],lambda=infinity,1);
series(tdL[1,2],lambda=infinity,1);
series(tdL[2,1],lambda=infinity,1);
```

$$\eta 0 := (q-t)\,t12 + t11 \qquad \textbf{(2.2)}$$

$$\left[\left[\frac{\lambda^2\,t12 + (-t\,t12 + t11)\,\lambda - q\,(q-t)\,t12 + (-p-t11)\,q + p\,t}{t-\lambda}, \frac{-\lambda+q}{t-\lambda}\right],\right.$$

$$\left[\frac{1}{(\lambda-t)\,(q-t)}\Big((-t12+t22)\,(q\,t12+p+t11)\,t^3+\big((3\,t12^2-2\,t12\,t22)\,q^2+(\lambda\,t12^2+(-\lambda\,t22+4\,p+4\,t11)\,t12-2\,t22\,(t11+p))\,q+((t11+p)\,\lambda-s10-s20-t20)\,t12-t22\,(t11+p)\,\lambda+p^2+2\,p\,t11-t10\,t22+t11^2\big)\,t^2+\big(t12\,(t22-3\,t12)\,q^3+(-2\,\lambda\,t12^2+(2\,\lambda\,t22-5\,p-5\,t11)\,t12+t22\,(t11+p))\,q^2+(((-2\,p-2\,t11)\,\lambda+2\,s10+2\,s20+t20)\,t12+2\,t22\,(t11+p)\,\lambda-2\,p^2-4\,p\,t11+t10\,t22-2\,t11^2)\,q+\lambda\,(s10+s20+t20)\,t12+t22\,\lambda\,t10+(s10+s20)\,(t11+p)\big)\,t+q^4\,t12^2+2\left(-\frac{1}{2}\,t22\,\lambda+\frac{1}{2}\,t12\,\lambda+p+t11\right)t12\,q^3+(((t11+p)\,\lambda-s10-s20)\,t12+(t11+p)\,(-\lambda\,t22+p+t11))\,q^2+(-\lambda\,(s10+s20+t20)\,t12-t22\,\lambda\,t10-(s10+s20)\,(t11+p))\,q+s10\,s20\Big),\ \frac{1}{\lambda-t}\Big((\lambda\,t22+q\,t12+p+t11+t21)\,t-\lambda^2\,t22-\lambda\,t21-q^2\,t12+(-p-t11)\,q+s10+s20\Big)\right]$$

$$-t12\,\lambda-t11+\mathrm{O}\left(\frac{1}{\lambda}\right)$$

$$1+\mathrm{O}\left(\frac{1}{\lambda}\right)$$

$$\frac{1}{q-t}\Big(((t12^2-t12\,t22)\,q+(t11+p)\,t12-t22\,(t11+p))\,t^2+((-2\,t12^2+2\,t12\,t22)\,q^2+((-2\,p-2\,t11)\,t12+2\,t22\,(t11+p))\,q+(s10+s20+t20)\,t12+t10\,t22)\,t+2\left(-\frac{1}{2}\,t22+\frac{1}{2}\,t12\right)t12\,q^3+((t11+p)\,t12-t22\,(t11+p))\,q^2+(-(s10+s20+t20)\,t12-t10\,t22)\,q\Big)+\mathrm{O}\left(\frac{1}{\lambda}\right)$$

Expression of the Lax matrices after symplectic reduction

```
> s20:=-s10-t10-t20:
  t21:=-t11:
  t22:=-t12:
  t11:=0:
```

```
t12:=1:
alpha12:=0:
alpha22:=0:
alpha11:=0:
alpha21:=-alpha11:
Lt:=1:
mu:=mubis;
nuMinus1:=nuMinus1;
c1:=c1bis;
c2:=c2bis;
c0:=0:

checkL:=simplify(checkL):
tdL:=simplify(tdL):

dG1dt:=Matrix(2,2,0):
for i from 1 to 2 do for j from 1 to 2 do dG1dt[i,j]:=diff(G1
[i,j],t)+diff(G1[i,j],q)*dqdt+diff(G1[i,j],p)*dpdt : od: od:

dqdt:=Lq/h:
dpdt:=Lp/h:
q:=checkq:
p:=checkp:
tdA:=simplify(Multiply(Multiply(G1,checkA),G1^(-1))+h*Multiply
(dG1dt,G1^(-1))):

checkL:=simplify(checkL);
tdL:=simplify(tdL);
checkA:=simplify(checkA);
tdA:=simplify(tdA);
```

$$\mu := \frac{q-t}{h} \tag{3.1}$$

$$nuMinus1 := 0$$

$$c1 := 0$$

$$c2 := 0$$

$$\left[\left[\frac{checkp\,(checkq-t)}{\lambda-t}, \frac{-\lambda+checkq}{t-\lambda}\right],\right.$$

$$\left[\frac{1}{(t-\lambda)\,(-checkq+t)}\Big((-checkq-\lambda)\,t^3+\big(checkp^2+2\,checkq^2+3\,checkq\,\lambda\right.$$

$$+2\,\lambda^2-h+t10-t20\big)\,t^2+\big(-checkq^3-3\,checkq^2\,\lambda+\big(-2\,checkp^2-3\,\lambda^2+h-t10$$

$+ t20) \, checkq - \lambda^3 + (t20 + h - t10) \, \lambda - (t10 + t20) \, checkp \big) \, t + \lambda \, checkq^3$

$+ \left(checkp^2 + \lambda^2\right) checkq^2 + \left(\lambda^3 + (-h + t10 - t20) \, \lambda + (t10 + t20) \, checkp\right) checkq$

$- s10 \, (s10 + t10 + t20) \Big), \dfrac{checkp \, (checkq - t) + t10 + t20}{t - \lambda} \Big] \Big]$

$\Big[\Big[\dfrac{(-\lambda + checkp + checkq) \, t - checkp \, checkq - checkq^2 + \lambda^2}{t - \lambda}, \dfrac{-\lambda + checkq}{t - \lambda} \Big],$

$\Big[\dfrac{1}{(t - \lambda) \, (-checkq + t)} \big(checkq^4 + (2 \, checkp + 2 \, \lambda - 4 \, t) \, checkq^3 + \big(5 \, t^2 + ($

$-6 \, checkp - 4 \, \lambda) \, t + checkp^2 + 2 \, checkp \, \lambda + t10 + t20 \big) \, checkq^2 + \big(-2 \, t^3 + (6 \, checkp$

$+ 2 \, \lambda) \, t^2 + \left(-2 \, checkp^2 - 4 \, checkp \, \lambda - 3 \, t10 - t20\right) t + (t10 + t20) \, checkp$

$+ 2 \, \lambda \, t10 \big) \, checkq - 2 \, checkp \, t^3 + \left(checkp^2 + 2 \, checkp \, \lambda + 2 \, t10\right) t^2 + \big((-t10$

$- t20) \, checkp - 2 \, \lambda \, t10 \big) \, t - s10 \, (s10 + t10 + t20) \big),$

$\dfrac{(-checkp - checkq + \lambda) \, t + checkp \, checkq + checkq^2 - \lambda^2 + t10 + t20}{t - \lambda} \Big] \Big]$

$\Big[\Big[-\dfrac{checkp \, (checkq - t)}{h \, (\lambda - t)}, \dfrac{checkq - t}{h \, (\lambda - t)} \Big],$

$\Big[\dfrac{1}{(-checkq + t) \, h \, (t - \lambda)} \big(t^4 + (-4 \, checkq - 2 \, \lambda) \, t^3 + \big(-checkp^2 + 5 \, checkq^2$

$+ 6 \, checkq \, \lambda + \lambda^2 \big) \, t^2 + \big(-2 \, checkq^3 - 6 \, checkq^2 \, \lambda + \left(2 \, checkp^2 - 2 \, \lambda^2\right) checkq$

$- \left(h^2 - h - t10 - t20\right) checkp \big) \, t + 2 \, \lambda \, checkq^3 + \left(-checkp^2 + \lambda^2\right) checkq^2 + \big(h^2$

$- h - t10 - t20 \big) \, checkp \, checkq + s10 \, (s10 + t10 + t20) \big),$

$\dfrac{h^2 - h + (-checkq + t) \, checkp - t10 - t20}{h \, (t - \lambda)} \Big] \Big]$

$\Big[\Big[\dfrac{(-checkp - \lambda - checkq + t) \, (-checkq + t)}{h \, (t - \lambda)}, \dfrac{checkq - t}{h \, (\lambda - t)} \Big],$

$\Big[\dfrac{1}{(-checkq + t) \, h \, (t - \lambda)} \big(-checkq^4 + (-2 \, checkp + 2 \, t) \, checkq^3 + \big(-checkp^2$

$+ 4 \, checkp \, t + h^2 - t^2 - h - t10 - t20 \big) \, checkq^2 - (-checkp + t) \, \big(2 \, checkp \, t + h^2 - h$

$$- t10 - t20)\ checkq - checkp^2\, t^2 - (h^2 - h - t10 - t20)\ checkp\, t + s10\ (s10 + t10$$

$$+ t20)),\ \frac{1}{h\,(t-\lambda)} \left(-t^2 + (checkp + 2\, checkq + \lambda)\, t - checkq^2 + (-checkp\right.$$

$$\left. - \lambda)\ checkq + h^2 - h - t10 - t20\right)\Big]\Big]$$